BANKISEL  -  генерация кода выбора банка RAM для косвенной адресации 


bankisel <label> 


Используется при создании объектного модуля. Эта директива представляет собой указание линкеру для генерации кода выбора соответствующего банка при косвенной адресации адреса назначенного <label>. Только одна метка <label> должна быть указана, над ней не должно выполняться никаких операций и она должна быть предварительно определена. 


Линкер сгенерирует код выбора соответствующего банка. Для устройств с14-битным ядром будет сгенерирована инструкция установки/сброса соответствующего бита RP в регистре STATUS. Для устройств с 16-битным ядром будет сгенерировано MOVLB или MOVLR. Если косвенный адрес полностью определен без этих инструкций не будет генерироваться никакого кода. For more information, refer to Chapter 4, "Using MPASM to Create Relocatable Objects." 


Example


 movlw	Var1


 movwf	FSR


 bankisel	Var1


 ...


 movwf	INDF





See Also    PAGESEL     BANKSEL





BANKSEL  - генерация кода выбора банка RAM


banksel <label> 


Используется при создании объектного модуля. Эта директива представляет собой указание линкеру для генерации кода выбора банка для установки банка, в котором находится адрес на который указывает <label>. Только одна метка <label> должна быть указана, над ней не должно выполняться никаких операций и она должна быть предварительно определена.


Линкер сгенерирует код выбора соответствующего банка. Для устройств с12-битным ядром будет сгенерирована инструкция установки/сброса соответствующего бита в регистре FSR.  Для устройств с14-битным ядром будет сгенерирована инструкция установки/сброса соответствующего бита RP в регистре STATUS. Для устройств с 16-битным ядром будет сгенерировано MOVLB или MOVLR. Если устройство содержит только один банк RAM, не будет генерироваться никакого кода. For more information, refer to Chapter 4, "Using MPASM to Create Relocatable Objects." 


Example


 banksel  Var1


 movwf    Var1





See Also     PAGESEL     BANKISEL





CBLOCK  -  определить группу констант


cblock  [<expr>] <label> [:<increment>][,<label>[:<increment>]]


	endc





Определяет список поименованных констант. Каждой метке <label> присваивается значение на единицу больше чем предыдущей. Эта директива служит для назначения смещения относительно начального адреса нескольким меткам. Список меток заканчивается, когда встречается директива ENDC. <expr>  указывает стартовое значение для первой метки в блоке. Если <expr> не указано, первая метка будет иметь значение на единицу больше чем последняя метка в предшествовавшей директиве CBLOCK. Если первая директива CBLOCK не имеет <expr>, присваивается начальное значение нуль. Если значение  <increment> определено, следующей метке <label> присваивается значение на <increment> больше чем предыдущей. Несколько имен, разделенных запятыми, может 


быть задано в одной строке. 


Example


cblock	  0x20              ; name_1 will be 


		                    ; assigned 20


		name_1, name_2		; name_2, 21 and so on


		name_3, name_4		; name_4 is assigned 23.


endc


cblock	  0x30


		TwoByteVar: 0, TwoByteHigh, TwoByteLow


		Queue: QUEUE-SIZE


		QueueHeadm QueueTail





Double1:2, Double2:2


endc





See Also     ENDC








CODE  -  начало секции кода в объектном модуле


[<label>] code [<ROM address>]





Используется только при создании объектных модулей. Объявляет начало секции программного кода. Если <label> не указана, секция будет названа .code  Стартовый адрес устанавливается равным указанному значению или нулю если адрес не был указан. For more information, refer to Chapter 4, "Using MPASM to Create Relocatable Objects."


Example


RESET   code    H'01FF'


        goto    START





See Also     IDATA    UDATA     UDATA_OVR     UDATA_SHR     EXTERN     GLOBAL





CONSTANT - объявление символьной константы 


constant <label>=<expr>  [...,<label>=<expr>]


Создает метки для использования в выражениях MPASM. Значение присвоенное этой директивой не может быть изменено после объявления и значение <expr> должно быть доступно во время присвоения. Это принципиальное отличие меток объявленных как CONSTANT от объявленных как VARIABLE или созданных директивой SET. Во всех остальных случаях константы и переменные равнозначны в выражениях. 


Example


variable	    RecLength=64	       ; Set Default 																						


					            ; RecLength


constant      BufLength=512	       ; Init BufLength


		.				       ; RecLength may


		.				       ; be reset later


		.				       ; in RecLength=128 


		.				       ;   


		.				       ;


constant MaxMem=RecLength+BufLength  ;CalcMaxMem





See Also        SET      VARIABLE


DA  -  сохранить строки в Program Memory


[<label>] da <expr> [, <expr2>, ..., <exprn>]





Создает 14-битное представление двух 7-битных ASCII символов. Это удобно для сохранения строк в памяти устройств с Flash ROM. 


Example


da "abcdef"     ;will put 30E2 31E4 32E6 3380 into program memory


da "12345678",0 ;will put 18B2 19B4 1AB6 0000 into program memory


da 0xFFFF       ;will put 0x3FFF into program memory





DATA - создание числовых и символьных данных 


[<label>]	data	<expr>, [,<expr>,...,<expr>]


[<label>]	data	"<text_string>" [,"<text_string>",...]





Инициализирует данные в одном или нескольких словах программной памяти. Данные могут быть константами, перемещаемыми или внешними метками или выражением, включающим вышеперечисленное. Данные также могут состоять из символов ASCII, <text_string>, одиночный символ заключается в одинарные кавычки, а строка в двойные. Одиночные символы помещаются в младший байт слова, а строки -- по два символа в слово, при этом первый символ помещается в старший байт. Если в строке нечетное число символов, в последний байт будет записан нуль. Когда создается объектный модуль эта директива также может быть использована для объявления инициализированных данных. Refer to the IDATA directive for more information. 


Example


data	  reloc_label+10	; constants


data	  1,2,ext_label	; constants, externals


data	  "testing 1,2,3"	; text string


data	  'N'			; single character


data	  start_of_program	; relocatable label





See Also      DW    DB    DE     DT      IDATA





DB - объявить байт данных


[<label>]  db  <expr>[,<expr>,...,<expr>] 





Резервирует программную память для 8-битных величин. Если объявлено несколько значений, то они последовательно заполняют байты. Если число значений нечетное, то в  конце будет добавлен нулевой байт. Когда создается объектный модуль, эта директива также может быть использована для объявления инициализированных данных. Refer to the IDATA directive for more information. 


Example


db	't', 0x0f, 'e', 0x0f, 's', 0x0f, 't', '\n'





See Also              DATA    DW    DT    DE    IDATA





DE - объявить байт данных в EEPROM 


[<label>]  de  <expr> [, <expr>, ..., <expr>]





Резервирует память для 8-битных данных. Каждое значение <expr> должно выражаться 8-битной величиной. Старшие биты программного слова обнуляются. Каждый символ в строке сохраняется как отдельное слово. Хотя директива была предназначена для инициализации EEPROM в PIC16C8X, она может быть использована для любого контроллера. 


Example


org	H'2100'		        ; Initialize EEPROM Data


de	"My Program, v1.0", 0





See Also        DATA      DB       DT       DW





DT - определить таблицу  


[<label>]  dt  <expr> [, <expr>, ..., <expr>]





Порождает последовательность инструкций RETLW, одна инструкция для каждого <expr>. Каждое <expr> должно быть 8-битной величиной. Каждый символ в строке сохраняется как операнд отдельной инструкции RETLW. 


Example


dt	"A Message", 0


dt	FirstValue, SecondValue, EndOfValues





See Also         DATA     DB      DE      DW





DW - объявление данных в одно слово 


[<label>]	dw  <expr>[,<expr>,...,<expr>]





Резервирует слова програмной памяти для данных, инициализируя это пространство заданными значениями. Значения сохраняются в памяти последовательно и программный счетчик инкрементируется на единицу. Выражения могут быть строками символов и сохраняются так же, как для директивы DATA. Когда создается объектный модуль, эта директива также может быть использована для объявления инициализированных данных. Refer to the IDATA directive for more information. 


Example


dw     39, "diagnostic 39", (d_list*2+d_offset)


dw     diagbase-1





See Also      DATA	DB       IDATA





ELSE  -  начало альтернативного пути ассемблирования в IF


Syntax


else


Используется в сочетании с директивой IF, обеспечивая альтернативный путь ассемблирования кода для случая, когда условие IF не выполняется. Директива ELSE может использоваться в пределах обычного программного блока или макроса. 


Example


speed	macro	rate


		if	rate < 50


			dw	slow


		else


			dw	fast


		endif


		endm





See Also       IF      ENDIF





END  -  конец программного блока


Syntax


end





Указывает конец программы. 


Example


start


	.			; executable code


	.			;


	.			;


		end		; end of instructions





ENDC  -  конец группы констант


endc





ENDC завершает список CBLOCK





See Also     CBLOCK





ENDIF  -  конец блока ассемблируемого по условию


endif





Директива заканчивает блок ассемблируемый по условию. Директива ELSE может использоваться в пределах обычного программного блока или макроса.





See Also    ELSE   IF





ENDM  -  конец макроопределения


endm 





Заканчивает макроопределение начатое директивой MACRO. 


Example


make_table macro arg1, arg2


           dw arg1, 0        ; null terminate table name


           res arg2          ; reserve storage


           endm





See Also    MACRO  EXITM





ENDW - завершает цикл WHILE 


endw


ENDW завершает цикл WHILE. До тех пор пока условие заданное директивой WHILE остается истинным, код между директивами WHILE и ENDW будет ассемблироваться. This directive may be used inside a regular program block or macro.


Example


See the example for WHILE





See Also WHILE





EQU - определить константу 


<label>      equ	<expr>  





Метке <label> присваивается значение <expr> 


Example


four	equ	4	; assigned the numeric value of 4 


			; to label four                                                                                           





See Also    SET





EXITM  -  выход из макроса


exitm





Обозначает немедленный возврат из макрорасширения во время ассемблирования. Имеет то же действие, как встретившаяся директива ENDM.


Example





test macro filereg


   if filereg == 1 ; check for valid file


      exitm


   else


      error "bad file assignment"


   endif


endm





See Also    ENDM    MACRO





EXTERN  - объявление метки определенной в другом модуле


extern  <label> [, <label>...]





Используется при создании объектного модуля. Объявляет символическое имя для использования в текущем модуле, но которое определено, как глобальное в другом модуле. Объявление EXTERN должно включаться до использования <label>. Хотя бы одна метка должна быть задана в строке. Если <label> определена в текущем модуле MPASM will generate a duplicate label error. For more information, refer to Chapter 6. 


Example


extern   Function


...


call     Function





See Also   GLOBAL   IDATA   TEXT   UDATA   UDATA_ACS   UDATA_OVR   UDATA_SHR





FILL - заполняет заданным значением область памяти 


[<label>]	fill	<expr>, <count>  





Генерирует <count> экземпляров значения <expr>. Если значение <expr> заключено в круглые скобки, то оно может быть инструкцией ассемблера. 


Example


fill	0x1009, 5	                  ; fill with a constant                                                     


fill	(GOTO RESET_VECTOR), NEXT_BLOCK-$                                 





See Also    DW     ORG     DATA   





GLOBAL  -  экспорт метки 


global  <label> [, <label>...]





Используется при создании объектного модуля. Объявляет символическое имя, которое определено в текущем модуле и доступно в других модулях. Объявление EXTERN должно включаться после определения <label>. Хотя бы одна метка должна быть задана в строке.


Example


         udata


Var1     res    1


Var2     res    1


         global Var1, Var2


         code


AddThree


         global AddThree


         addlw  3


         return





See Also   EXTERN   IDATA   TEXT   UDATA   UDATA_ACS    UDATA_OVR   UDATA_SHR





IDATA  -  начало секции инициализированных данных в объектном модуле


[<label>]  idata  [<RAM address>]





Используется при создании объектного модуля. Объявляет начало секции инициализирован- ных данных. Если метка <label> не указана секция будет названа .idata Стартовый адрес инициализируется с указанного адреса или назначается в процессе линковки, если адрес не был указан. Код не может быть помещен в эту область. Линкер сгенерирует список с записями для каждого байта указанного в разделе idata. The user must then link or include the appropriate initialization code. See Chapter 9 for examples of initialization codes for various PICmicro families. Note that this directive is not available for 12-bit core devices. 


Директивы RES, DB и DW могут быть использованы для резервирования пространства под переменные. RES создает инициализирующее значение нуль. DB инициализирует последовательно расположенные байты RAM. DW инициализирует последовательно расположенные байты RAM в порядке: младший байт/старший байт. For more information, refer to Chapter 6. 


Example


       idata


LimitL dw   0


LimitH dw D'300'


Gain   dw D'5'


Flags  db 0


String db 'Hi there!'





See Also  EXTERN  GLOBAL   TEXT   UDATA   UDATA_ACS   UDATA_OVR   UDATA_SHR





IF  -  начало секции кода ассемблируемого по условию


if <expr> 





Начинает секцию кода ассемблируемого по условию. Если <expr> является истинным, то код непосредственно следующий за IF будет ассемблирован. В противном случае код между IF и ближайшей директивой ELSE или ENDIF ассемблироваться не будет. Значение <expr>


равное нулю считается logically FALSE, любое другое значение считается logically TRUE. Директивы IF и WHILE используют логические значения выражения. Соответственно TRUE выражения возвращают ненулевое значение, FALSE - нулевое. 


Example


if version == 100; check current version


   movlw 0x0a


   movwf io_1


else


   movlw 0x01a


   movwf io_2


endif


See Also    ELSE    ENDIF





IFDEF - выполнить, если значение метки было задано


ifdef  <label> 





Если метка <label> была предварительно определена, обычно в результате применения директивы #define или установки значения в командной строке MPASM, то выполняется ветвь условия. Ассемблирование будет продолжаться, пока не встретится закрывающая директива ELSE или ENDIF. 


Example


#define	testing	1	    ; set testing "on"


		.


		.


		.                           						


ifdef	testing                                                                                    


	<execute test code>	    ; this path would                     


endif				    ; be executed.





See Also    #DEFINE    ELSE    ENDIF    IFNDEF     #UNDEFINE








IFNDEF - выполнить, если значение метки не было задано 


ifndef  <label> 





Если значение метки <label> не было предварительно задано или было удалено в результате применения директивы #UNDEFINE то код, следующий за директивой, будет ассемблирован. Ассемблирование будет разрешено или запрещено до тех пор пока не встретится закрывающая директива ELSE или ENDIF. 


Example


#define	testing1	     ; set testing on


	.


	.


	.


#undefine 	testing1	; set testing off


	ifndef	testing1	; if not in testing mode


     .			     ; execute


	.			     ; this path          


	.			     ; 					                


	endif			;


			          ;                                                           


     end		          ; end of source  





See Also     #DEFINE     ELSE     IFDEF   #UNDEF    INE     ENDIF





INCLUDE - включение дополнительного Source File


include <<include_file>>


include "<include_file>"





Укаазанный файл читается как исходник. Эффект тот же самый, как если бы целый текст включаемого файла был вставлен в месте размещения директивы INCLUDE. По окончании включаемого файла будет продолжаться ассемблирование исходника. Допускается до шести уровней вложения. <include_file> может быть заключен в кавычки или угловые скобки. Если указан полный путь к файлу, то поиск будет происходить только по этому пути. В противном случае порядок поиска следующий: текущий рабочий каталог, каталог в котором находится исходник, каталог MPASM. 


Example


include "c:\sys\sysdefs.inc" ; system defs


include <regs.h>             ; register defs





LOCAL  -  объявление локальной макропеременной


local <label>[,<label>...]


Объявляет указанные элементы данных, которые действительны в локальном контексте макроса. <label> может быть такой же как другая метка, объявленная вне макроопределения; это не будет вызывать конфликт между ними. Если макрос вызывается рекурсивно то каждое вхождение будет иметь собственную копию локальных данных. 


Example





<main code segment>


      :


      :


len   equ 10           ; global version


size  equ 20           ; note that a local variable


                       ; may now be created and modified


test macro size        ;


      local len, label ; local len and label


len   set size         ; modify local len


label res len          ; reserve buffer


len   set len-20       ;


endm                   ; end macro





See Also    ENDM    MACRO





MACRO  -  объявление макроопределения


<label> macro [<arg>, ..., <arg>]





Макрос - последовательность инструкций, которые могут быть вставлены в исходник используя единственный вызов макроса. Макрос сперва должен быть определен, затем в последующем коде возможны ссылки на него. Макрос может вызывать другой макрос или вызывать сам себя (рекурсия). Please refer to Chapter 5, "Macro Language" for more information.


Example


Read macro device, buffer, count


     movlw device


     movwf ram_20


     movlw buffer   ; buffer address


     movwf ram_21


     movlw count    ; byte count


     call  sys_21   ; read file call


endm





See Also    ELSE    ENDIF    ENDM    EXITM    IF    LOCAL





ORG - установка адреса начала программы


[<label>]  org   <expr>





Устанавливает начало программы для последующего кода с адреса определенного в значении <expr>. Если задана метка <label> то она будет иметь значение <expr>. Если ORG не задана, код будет формироваться с нулевого адреса. Директива может не использоваться если создается объектный модуль. 


Example


int_1	org	0x20	        ; Vector 20 code goes here 


int_2	org	int_1+0x10   ; Vector 30 code goes here 		                               





See Also    RES       FILL





PAGESEL  -  генерация кода выбора страницы программной памяти


pagesel  <label>





Используется при создании объектного модуля. Директива представляет собой указание линкеру генерировать код устанавливающий биты выбора страницы для установки страницы содержащей место, которому назначена <label>. Только одна метка <label> должна быть указана, над ней не должно выполняться никаких операций и она должна быть предварительно определена. Линкер сгенерирует код выбора соответствующей страницы. Для устройств с 12-битным ядром генерируется инструкция установки/сброса соответствую- щего бита в регистре STATUS. Для устройств с 14- и 16-битным ядром генерируются инструкции MOVLW и MOVWF для модификации PCLATH. Если устройство содержит только одну страницу программной памяти, не будет генерироваться никакого кода. For PIC18CXX devices, this command will do nothing. For more information, refer to Chapter 6.


Example


pagesel GotoDest


goto    GotoDest


....


pagesel CallDest


call    CallDest





See Also   BANKISEL    BANKSEL





PROCESSOR  -  установка типа контроллера


processor <processor_type>





Устанавливает указанный тип контроллера. 


Example


processor 16C54





See Also  LIST





RADIX  -  устанавливает основание системы исчисления по умолчанию


radix <default_radix>





Устанавливает default radix для выражений содержащих данные. The default radix is hex. Valid radix values are: hex, dec, or oct. 


Example


radix dec





See Also   LIST





RES - зарезервировать память 


[<label>]	res	<mem_units>





Увеличивает текущее значение программного счетчика на величину, заданную в <mem_units>. Следует обратить внимание, что <label> будет обозначать адрес, а не константу или переменную. В неперемещаемом коде  <label> ассоциируется с адресом памяти программ. В перемещаемом коде директива res также может быть использована для резервирования памяти данных. Расположение адреса определено в словах для 12-, 14- and 16-bit PICmicros, и в байтах для enhanced 16-bit PICmicros.


Example                                                                                                                          


buffer  res  64	; reserve 64 words of storage    				          


See Also     ORG        FILL





SET - определить переменную 


<label>  set  <expr> 





Директива SET эквивалентна директиве EQU за исключением того, что значение <expr> присвоенное метке этой директивой, впоследствии может быть изменено другой директивой SET. 


Example


area	set	0


width	set	0x12


length	set	0x14


area	set	length * width 					                                            


length	set	length + 1 					                                      





See Also    EQU     VARIABLE





VARIABLE - объявление символьной переменной  


variable <label>[=<expr>][,<label>[=<expr>]...]





Создает символические имена для использования в выражениях MPASM. Переменные и константы взаимозаменяемы в выражениях. Символические имена создаваемые директивой VARIABLE и создаваемые директивой SET, функционально эквивалентны. Разница состоит в том, что директива VARIABLE не требует, чтобы символы инициализировались при объявлении. Значение переменной не может быть изменено внутри операнда. Присвоения, инкремент и декремент помещаются в отдельную строку. 


Example  


Please refer to the example given for the CONSTANT directive.


See Also    CONSTANT    SET





WHILE  -  выполнять цикл пока условие остается истинным


while <expr>


:


:


endw





Строки между WHILE и ENDW ассемблируются до тех пор пока <expr> остается истинным. Выражение с нулевым значением считается logically FALSE. Выражение с любым другим значением считается logically TRUE. Выражение со значением TRUE возвращает ненулевое значение, выражение со значением FALSE возвращает нулевое значение. Цикл WHILE может содержатьне более 100 строк и повторяется максимум 256 раз. 








Example


test_mac macro count


         variable i


i = 0


         while  i  < count


         movlw  i


i += 1


         endw


         endm


start


         test_mac  5


         end





See Also   ENDW    IF





#DEFINE - задает строку, которой будет замещена метка 


#define <name> [<string>] 





Директива задает строку <string>, замещающую метку <name> всякий раз, когда та будет встречаться в исходном тексте. Использование директивы без параметра <string> causes a definition of <name> to be noted internally and may be tested for using the IFDEF directive. 


Директива подражает стандарту ANSI 'C ' для #define. Символы заданные этим способом недоступны для просмотра в MPLAB. 


Example


#define	length	20


#define	control	0x19,7


#define	position     (X,Y,Z)     (Y-(2 * Z +X))


	.


	.


	.


test_label  dw  position(1, length, 512)


	bsf	control	                   ; set bit 7 in f19                                                           





See Also  IFDEF    IFNDEF    #UNDEFINE 





#UNDEFINE  -  удаление подставляемой метки


#undefine <label>





<label> это идентификатор предварительно определенный директивой #DEFINE. Это должна быть действительная для MPASM метка. Директива #UNDEFINE удаляет символическое имя из symbol table.


Example


#define   length 20


:


:


#undefine length





See Also   #DEFINE    IFDEF   INCLUDE   IFNDEF





_ _CONFIG  - установка битов конфигурации контроллера


_ _config <expr> OR _ _config <addr>, <expr>





Устанавливает биты конфигурации контроллера в значение <expr>. Для PIC18CXX адрес байта конфигурации должен быть указан в <addr>. Refer to the PICmicro Microcontroller Data Book for a description of the configuration bits for each processor. Перед использованием этой директивы контроллер должен быть объявлен в командной строке, директиве LIST или директиве PROCESSOR. Если эта директива используется для PIC17CXX family, выходной формат hex файла должен быть установлен INHX32 в командной строке или директивой LIST.


Example





list p=17c42,f=INHX32


_ _config H'FFFF'      ;Default configuration b





See Also    _ _IDLOCS    LIST    PROCESSOR





_ _BADRAM - устанавливает неиспользуемые адреса RAM 


_ _MAXRAM - устанавливает максимальный адрес RAM





Директивы _ _MAXRAM и _ _BADRAM вместе отмечают доступность неиспользуемых регистров. _ _BADRAM отмечает неиспользуемые адреса RAM. Эта директива используется только вместе с директивой _ _MAXRAM, которая должна предварять _ _BADRAM. Каждое значение <expr> должно быть меньше или равно значению определяемому _ _MAXRAM. Когда используется директива _ _MAXRAM непосредственно указываемые адреса RAM проверяются на доступность по карте RAM, создаваемой директивой _ _BADRAM. 


_ _MAXRAM определяет максимальный доступный адрес RAM и инициализирует карту доступности для всех адресов, начиная от значения <expr> и ниже. Значение <expr> должно быть больше или равно максимальному адресу RAM в банке 0 и меньше чем 1000h. В исходном файле директива _ _MAXRAM может быть использована несколько раз. Каждое использование переопределяет значение максимально доступного адреса RAM и переустанавливает карту доступных адресов RAM. 


Example


list	p=16c622


_ _maxram	H'0BF'


_ _badram	H'07'-H'09', H'0D'-H'1E'


_ _badram	H'87'-H'89', H'8D', H'8F'-H'9E'


movwf	H'07'	    ; Generates invalid RAM warning


movwf	H'87'	    ; Generates invalid RAM warning


		              ;   and truncation message     








Expressions


Expressions are used in the operand field of the source line and may contain constants, symbols, or any combination of constants and symbols separated by arithmetic operators. Each constant or symbol may be preceded by a plus or minus to indicate a positive or negative expression.





Symbol


A symbol is a general purpose mechanism for describing the various pieces which comprise a program. These pieces include function names, variable names, section names, file names, struct/enum/union tag names, etc. 








Директивы ассемблера распологаются в тексте исходного файла для расширения функций ассемблера,  определения и управлением процессом. Общее правило - директивы не должны располагаться в первой позиции строки в тексте (иначе будут определены как метка)

